簿記原理とコンピュータ

武 田 嘉 孝

Bookkeeping Principles and the Computer

Yoshitaka Takeda

Summary

In my previous paper I introduced "Computerising Journal Entries".The paper explained how to keep transactions in Bookkeeping through Cobol Programming. This field is in the category of Accounting. The definition of accounting is *recognizing, measuring* and *keeping*, and *reporting* the results of business economic activities. These actions are called *Accounting Actions*. Accounting Actions are constituted of Accounting Processing, Keeping and Reporting. In Acconuting Actions, Keeping is in the category of Bookkeeping Principle. For many businesses, the computerisation of accounts is becoming one of the most commonplace ways in which high technology is being introduced for the benefit of productivity. Therefore, we are investigating Computerisation of Accounts in Bookkeeping Principles for the transition from a manual accounting system to a computer accounting system.

Received Apr.30,1996

Key words: Trial Balance (TB)
Profit and Loss Statement (PL)
Balance Sheet (BS)
Closing Trial Balance (KURITB)

武 田 嘉 孝

目　　　次

は じ め に

簿記原理は会計学総論において重要な基本的原理である。それは次の会計の定義によって明らかとなる。会計とは，企業の経済活動の結果を認識，測定，記録しその結果を報告することである。この内，認識・測定は会計処理と呼ばれ，記録，報告と並んで「広義の会計」といわれる。会計処理は「狭義の会計」である。認識とは簿記上の取引を資産，負債，資本，収益，費用という五つの勘定のうち，どの勘定科目で処理をするかを決めることであり，また測定とはその取引の内容をいついくらの金額でそれを計上するかを決めることである。こうした「狭義の会計」は会計学原理で取り扱われる。次に記録とは帳簿記録すなわち“Book”を帳簿といい，“Keeping”は記録（記入）という。これらのことから Bookkeeping という英米の文献の専門用語を用いて簿記と呼んでいる。

簿記の対象となるいわゆる簿記上の取引とは，その経済主体（ここでは企業）の財産に変動が生じた事柄を言う。その取引は，簿記原理の拠り所となっている複式簿記の法則に基づき，具体的な勘定科目と金額を借方と貸方とに分けて記入される。これを仕訳と呼ぶ。仕訳は，取引の内容の金額が借方と貸方に分けて二面的に把握される。このことは，複式簿記のいわゆる貸借平均の原則が働いていることになる。一つの取引の金額が借方と貸方とに二面的に把握されることになることから，複式というのは，すなわち“Double-keeping”複式記入という意味をもっている。簿記原理で取り扱われる帳簿には，仕訳を記入する仕訳帳や伝票がある。この帳簿は取引の内容を歴史的に記録することを目的として作成され，企業の取引の実態を明らかにするための帳簿である。仕訳帳で記入された取引を勘定科目別にその増減を明らかにする帳簿を総勘定元帳という。仕訳帳と総勘定元帳はいずれも企業の経済活動の動きを把握するうえで欠くことのできない帳簿であるところから，主要簿と呼ばれている。この主要簿の明細を記録するものを補助簿といい，それはさらに補助元帳と補助記入帳とに分かれている。このように主要簿と補助簿とを，複式簿記機構上の帳簿すなわち会計帳簿という。会計では，これらの帳簿以外に決算帳簿といわれるものがある。一般に会計報告書または財務諸表と呼ばれるものがこれである。先述した会計の定義では，会計処理，記録，報告（伝達）というものがあった。このように会計の定義に示された会計処理および記録さらに報告の三つの行為は会計行為という。会計処理では「狭義の会計」で検討されたものを，複式簿記の考え方に基づいて主要簿と補助簿に記録され，決算の結果財務諸表が作成される。簿記原理ではこのような帳簿を記録・作成することが取り扱われる。よって本稿では，入力データから合計残高試算表，損益計算書，貸借対照表，繰越試算表を作成するための手続すなわち会計サイクルをコンピュータ・プログラムで実行する方法を提示することとしたい。

I 合計残高試算表

(1) 入力データ

① SIWAKEA.DAT

```
B:¥SIWAKEA.DAT                         [    1:   1]      <挿入>
----+----1----+----2----+----3----+----4----+----5----+----6----+-
080301000111119999000056850
080301000111319999000044600
080301000121519999000189960
080301000121619999000024420
080301000199994120000012220
080301000199997110000303610
080302000211118111000015810
080305000383214120000030000
080306000421511111000050000
080310000511318111000014800
080313000641201111000008000
080317000711111131000025000
080320000883161111000002900
080320000983134120000003900
080321001011318111000029200
080323001183151111000018000
080325001211111131000012000
080329001383161111000003805
080329001441201111000025000
080330001511118111000024200
080331001683171111000003000
   1ﾒﾆｭｰ1  2ﾒﾆｭｰ2  3ﾒﾆｭｰ3  4画面割5検索↓   6行選択7カット8コピー9行へ
```

② KANJO.DAT

```
1111現金
1131売掛金
1210商品
2131機械
2151車両運搬具
2152自動車
2161備品
2162図書資料
4120買掛金
4130借入金
7110資本金
8111売上
8311運搬費
8312電力料
8313自動車費
8314用水費
8315給料
8316雑費
8317家賃
8318消耗品費
8319広告料
8320通信費
8321燃料費
8322営業雑費
8323諸料金
8810支払利息
9998当期純損失
9999当期純利益
```

(2) ソースプログラム

```
IDENTIFICATION        DIVISION.
PROGRAM-ID.           TB.
AUTHOR.               武田  嘉孝(12345).
ENVIRONMENT           DIVISION.
CONFIGURATION         SECTION.
SOURCE-COMPUTER.      PC-9801.
OBJECT-COMPUTER.      PC-9801.
SPECIAL-NAMES.        CONSOLE  IS  CRT.
INPUT-OUTPUT          SECTION.
FILE-CONTROL.
     SELECT SIWAKE-F  ASSIGN "B:SIWAKEA.DAT".
     SELECT KANNAME-F ASSIGN "B:KANJO.DAT".
     SELECT KESSAN-F  ASSIGN "B:KESSAN.DAT".
     SELECT INSATU-F  ASSIGN "PRN:".
DATA                  DIVISION.
FILE                  SECTION.
FD   KANNAME-F.
01   KANNAME-REC.
     02   FILLER      PIC X(22).
FD   SIWAKE-F.
01   SIWAKE-REC.
     02   I-NEN       PIC 9(2).
     02   I-TUKI      PIC 9(2).
     02   I-HI        PIC 9(2).
     02   I-BAN       PIC 9(4).
     02   I-KR        PIC 9(4).
     02   I-KS        PIC 9(4).
     02   I-KINGAKU   PIC 9(8).
     02   FILLER      PIC X(2).
FD   KESSAN-F.
01   KESSAN-REC.
     02   K-CODE      PIC 9(4).
     02   K-KR-ZAN    PIC 9(8).
     02   K-KS-ZAN    PIC 9(8).
     02   FILLER      PIC X(2).
FD   INSATU-F  LABEL  RECORD  OMITTED.
01   INSATU-REC.
     02   FILLER      PIC X(80).
WORKING-STORAGE       SECTION.
77   I                PIC 9(2).
77   W                PIC 9    VALUE  2.
77   WW               PIC 9(2).
77   G-KR-KIN         PIC 9(8) VALUE  0.
77   G-KS-KIN         PIC 9(8) VALUE  0.
77   G-KR-ZAN         PIC 9(8) VALUE  0.
77   G-KS-ZAN         PIC 9(8) VALUE  0.
77   W-EOF01          PIC 9    VALUE  0.
77   W-EOF02          PIC 9    VALUE  0.
01   MIDASI01.
     02   FILLER  PIC X(28) VALUE  SPACE.
     02   FILLER  PIC X(52) VALUE "合  計  残  高  試  算  表".
01   MIDASI02.
     02   FILLER  PIC X(32) VALUE  SPACE.
     02   O-NEN   PIC Z(2).
     02   FILLER  PIC X(4)  VALUE "年".
     02   O-TUKI  PIC Z(2).
     02   FILLER  PIC X(4)  VALUE "月".
     02   O-HI    PIC Z(2).
     02   FILLER  PIC X(34) VALUE "日".
01   MIDASI03.
```

```
       02   FILLER    PIC X(6)  VALUE   SPACE.
       02   FILLER    PIC X(27) VALUE "   借         方    ".
       02   FILLER    PIC X(23) VALUE "勘  定  科  目".
       02   FILLER    PIC X(24) VALUE "   貸            方  ".
01     MIDASI04.
       02   FILLER    PIC X(4)  VALUE   SPACE.
       02   FILLER    PIC X(16) VALUE "残     高".
       02   FILLER    PIC X(32) VALUE "合       計".
       02   FILLER    PIC X(20) VALUE "合       計".
       02   FILLER    PIC X(8)  VALUE "残     高".
01     OUT-R.
       02   FILLER    PIC X(2)  VALUE   SPACE.
       02   O-KR-ZAN  PIC ¥¥,¥¥¥,¥¥¥.
       02   FILLER    PIC X(8)  VALUE   SPACE.
       02   O-KR-KIN  PIC ¥¥,¥¥¥,¥¥¥.
       02   FILLER    PIC X(3)  VALUE   SPACE.
       02   O-NAME    PIC X(16).
       02   FILLER    PIC X(3)  VALUE   SPACE.
       02   O-KS-KIN  PIC ¥¥,¥¥¥,¥¥¥.
       02   FILLER    PIC X(8)  VALUE   SPACE.
       02   O-KS-ZAN  PIC ¥¥,¥¥¥,¥¥¥.
01     LINE01.
       02   FILLER    PIC X(2)  VALUE   SPACE.
       02   FILLER    PIC X(28) VALUE   ALL "－".
       02   FILLER    PIC X(22) VALUE   SPACE.
       02   FILLER    PIC X(28) VALUE   ALL "－".
01     LINE02.
       02   FILLER    PIC X(2)  VALUE   SPACE.
       02   FILLER    PIC X(28) VALUE   ALL "＝".
       02   FILLER    PIC X(22) VALUE   SPACE.
       02   FILLER    PIC X(28) VALUE   ALL "＝".
01     KANNAME          OCCURS  30.
       02   CODE        PIC 9(4).
       02   NAME        PIC X(16).
01     HOZON.
       02   HOZONA      OCCURS  30.
         03  H-CODE     PIC 9(4).
         03  H-KR-KIN   PIC 9(8).
         03  H-KS-KIN   PIC 9(8).
01     CRT-HYOUJI.
       02   CRT-NAME PIC X(80) VALUE "決算日を入力してください。".
       02   CRT-NEN  PIC X(80) VALUE "年を入力[   ]".
       02   CRT-TUKI PIC X(80) VALUE "月を入力[   ]".
       02   CRT-HI   PIC X(80) VALUE "日を入力[   ]".
01     ENTER-SYORI.
       02   FILLER    PIC X(80) VALUE   SPACE.
       02   FILLER    PIC X(9)  VALUE   SPACE.
       02   N-NEN     PIC 9(2).
       02   FILLER    PIC X(69) VALUE   SPACE.
       02   FILLER    PIC X(9)  VALUE   SPACE.
       02   N-TUKI    PIC 9(2).
       02   FILLER    PIC X(69) VALUE   SPACE.
       02   FILLER    PIC X(9)  VALUE   SPACE.
       02   N-HI      PIC 9(2).
       02   FILLER    PIC X(69) VALUE   SPACE.
PROCEDURE               DIVISION.
HAJIMARI.
       OPEN INPUT SIWAKE-F KANNAME-F OUTPUT KESSAN-F INSATU-F.
       READ  SIWAKE-F  AT  END  MOVE  1  TO  W-EOF01.
       PERFORM  SYORI01  UNTIL  W-EOF01  =  1.
OWARI.
       CLOSE  SIWAKE-F  KANNAME-F  KESSAN-F  INSATU-F.
```

```
     STOP   RUN.
SYORI01.
     PERFORM  NYURYOKU.
     PERFORM  KAN-NAME    VARYING  WW  FROM  1  BY  1
                          UNTIL  WW  >  30  OR  1  =  W-EOF02.
     PERFORM  SYORI02     UNTIL  W-EOF01  =  1.
     PERFORM  MIDASI-INZI.
     PERFORM  INZI        VARYING  I   FROM  1  BY  1
                          UNTIL  I   >  (WW  -  1).
     PERFORM  GOKEI-INZI.
SYORI02.
     PERFORM  SAGASI      VARYING  I   FROM  1  BY  1
                          UNTIL  I   >  (WW  -  1).
     READ  SIWAKE-F  AT  END  MOVE  1  TO  W-EOF01.
SAGASI.
     IF  H-CODE(I) =  I-KR  ADD  I-KINGAKU  TO  H-KR-KIN  (I).
     IF  H-CODE(I) =  I-KS  ADD  I-KINGAKU  TO  H-KS-KIN  (I).
MIDASI-INZI.
     WRITE    INSATU-REC  FROM  MIDASI01  AFTER  1.
     WRITE    INSATU-REC  FROM  MIDASI02  AFTER  3.
     WRITE    INSATU-REC  FROM  MIDASI03  AFTER  3.
     WRITE    INSATU-REC  FROM  MIDASI04  AFTER  2.
KAN-NAME.
     READ KANNAME-F INTO KANNAME (WW) AT END MOVE 1 TO W-EOF02.
     MOVE  CODE  (WW)  TO  H-CODE (WW).
GOKEI-INZI.
     MOVE   SPACE   TO   OUT-R.
     WRITE    INSATU-REC  FROM  LINE01   AFTER  1.
     MOVE   SPACE   TO   OUT-R.
     MOVE   "                    "  TO  O-NAME.
     MOVE   G-KR-KIN     TO  O-KR-KIN.
     MOVE   G-KS-KIN     TO  O-KS-KIN.
     MOVE   G-KS-ZAN     TO  O-KS-ZAN.
     MOVE   G-KR-ZAN     TO  O-KR-ZAN.
     WRITE    INSATU-REC  FROM  OUT-R    AFTER  1.
     WRITE    INSATU-REC  FROM  LINE02   AFTER  1.
INZI.
     MOVE   SPACE        TO  KESSAN-REC  OUT-R.
     MOVE   CODE  (I)    TO  K-CODE.
     MOVE   NAME  (I)    TO  O-NAME.
     IF H-KR-KIN  (I) >  H-KS-KIN  (I)
            COMPUTE K-KR-ZAN  =  H-KR-KIN(I) - H-KS-KIN(I)
            MOVE  K-KR-ZAN     TO  O-KR-ZAN
            MOVE  H-KR-KIN(I)  TO  O-KR-KIN
            MOVE  H-KS-KIN(I)  TO  O-KS-KIN
            ADD   H-KR-KIN(I)  TO  G-KR-KIN
            ADD   H-KS-KIN(I)  TO  G-KS-KIN
            ADD   K-KR-ZAN     TO  G-KR-ZAN
            WRITE   INSATU-REC  FROM  OUT-R    AFTER  W
            WRITE   KESSAN-REC  AFTER  1.
     IF H-KR-KIN  (I) <  H-KS-KIN  (I)
            COMPUTE K-KS-ZAN  =  H-KS-KIN(I) - H-KR-KIN(I)
            MOVE  K-KS-ZAN     TO  O-KS-ZAN
            MOVE  H-KR-KIN(I)  TO  O-KR-KIN
            MOVE  H-KS-KIN(I)  TO  O-KS-KIN
            ADD   H-KR-KIN(I)  TO  G-KR-KIN
            ADD   H-KS-KIN(I)  TO  G-KS-KIN
            ADD   K-KS-ZAN     TO  G-KS-ZAN
            WRITE  INSATU-REC  FROM  OUT-R    AFTER  W
            WRITE   KESSAN-REC  AFTER  1.
NYURYOKU.
     DISPLAY  SPACE  UPON  CRT.
```

```
002850      DISPLAY  CRT-HYOUJI  AT  0102  UPON  CRT.
002860      DISPLAY  ENTER-SYORI   AT  0102  UPON  CRT.
002870      ACCEPT   ENTER-SYORI   AT  0102  FROM  CRT.
002880      MOVE   N-NEN   TO   O-NEN.
002890      MOVE   N-TUKI  TO   O-TUKI.
002900      MOVE   N-HI    TO   O-HI.
```

(3) 実行結果

① 合計残高試算表

合 計 残 高 試 算 表

8年 3月 31日

借方 残高	借方 合計	勘定科目	貸方 合計	貸方 残高
¥23,155	¥133,860	現金	¥110,705	
¥51,600	¥88,600	売掛金	¥37,000	
¥239,960	¥239,960	車両運搬具		
¥24,420	¥24,420	備品		
	¥33,000	買掛金	¥46,120	¥13,120
		資本金	¥303,610	¥303,610
		売上	¥84,010	¥84,010
¥3,900	¥3,900	自動車費		
¥18,000	¥18,000	給料		
¥6,705	¥6,705	雑費		
¥3,000	¥3,000	家賃		
¥30,000	¥30,000	燃料費		
¥400,740	¥581,445		¥581,445	¥400,740

② 決算データ（貸借対照表，損益計算書の入力ファイルとして用いる）

```
B:¥KESSAN.DAT                         [    13:    1]      <挿 入>          HELP=ヘル
----+----1----+----2----+----3----+----4----+----5----+----6----+----7----+--
111100023155
113100051600
215100239960
216100024420
4120        00013120
7110        00303610
8111        00084010
831300003900
831500018000
831600006705
831700003000
832100030000
[EOF]
~
~
~
~
~
~
~
~
   1ﾒﾆｭｰ1  2ﾒﾆｭｰ2  3ﾒﾆｭｰ3  4画面割5検索↓    6行選択7カット8コピー9行ﾍﾟｽﾄ0タグJP
```

II 損益計算書

(1) ソースプログラム

```
001010 IDENTIFICATION        DIVISION.
001020 PROGRAM-ID.           PL.
001030 AUTHOR.               武田  嘉孝(12345).
001040 ENVIRONMENT           DIVISION.
001050 CONFIGURATION         SECTION.
001060 SOURCE-COMPUTER.      PC-9801.
001070 OBJECT-COMPUTER.      PC-9801.
001080 SPECIAL-NAMES.        CONSOLE  IS  CRT.
001090 INPUT-OUTPUT          SECTION.
001100 FILE-CONTROL.
001120     SELECT KANNAME-F  ASSIGN "B:KANJO.DAT".
001130     SELECT KESSAN-F   ASSIGN "B:KESSAN.DAT".
001140     SELECT INSATU-F   ASSIGN "PRN:".
001150 DATA                  DIVISION.
001160 FILE                  SECTION.
001170 FD  KANNAME-F.
001180 01  KANNAME-REC.
001190     02   FILLER       PIC X(22).
001200 FD  KESSAN-F.
001210 01  KESSAN-REC.
001220     02   K-CODE       PIC 9(4).
001230     02   K-KR-ZAN     PIC 9(8).
001240     02   K-KS-ZAN     PIC 9(8).
001250     02   FILLER       PIC X(2).
001260 FD  INSATU-F  LABEL  RECORD  OMITTED.
001270 01  INSATU-REC.
001280     02   FILLER       PIC X(80).
001290 WORKING-STORAGE       SECTION.
001300 77  I                 PIC 9(2).
001310 77  T                 PIC 9(2).
001320 77  WW                PIC 9(2).
001330 77  KR                PIC 9(2) VALUE  1.
001340 77  KS                PIC 9(2) VALUE  1.
001350 77  G-KR-KIN          PIC 9(8) VALUE  0.
001360 77  G-KS-KIN          PIC 9(8) VALUE  0.
001370 77  W-EOF01           PIC 9    VALUE  0.
001380 77  W-EOF02           PIC 9    VALUE  0.
001390 01  MIDASI01.
001400     02   FILLER    PIC X(30) VALUE   SPACE.
001410     02   FILLER    PIC X(50) VALUE "損  益   計   算   書".
001420 01  MIDASI02.
001430     02   FILLER    PIC X(19) VALUE   SPACE.
001440     02   O-NEN-01  PIC Z(2).
001450     02   FILLER    PIC X(4)  VALUE "年".
001460     02   O-TUKI-01 PIC Z(2).
001470     02   FILLER    PIC X(6)  VALUE "月".
001480     02   FILLER    PIC X(9)  VALUE "1日から".
001490     02   O-NEN-02  PIC Z(2).
001500     02   FILLER    PIC X(4)  VALUE "年".
001510     02   O-TUKI-02 PIC Z(2).
001520     02   FILLER    PIC X(4)  VALUE "月".
001530     02   O-HI-02   PIC Z(2).
001540     02   FILLER    PIC X(24) VALUE "日まで".
001550 01  MIDASI03.
001560     02   FILLER  PIC X(9)  VALUE   SPACE.
001570     02   FILLER  PIC X(30) VALUE   ALL  "─".
001580     02   FILLER  PIC X(2)  VALUE   "┬".
001590     02   FILLER  PIC X(30) VALUE   ALL  "─".
```

```
           02    FILLER    PIC X(9)   VALUE    SPACE.
       01  OUT-R.
           02    FILLER    PIC X(9)   VALUE    SPACE.
           02    O-KR      PIC X(16).
           02    FILLER    PIC X(3)   VALUE    SPACE.
           02    O-KR-KIN  PIC ¥¥¥,¥¥¥,¥¥¥.
           02    FILLER    PIC X(2)   VALUE    "|".
           02    O-KS      PIC X(16).
           02    FILLER    PIC X(3)   VALUE    SPACE.
           02    O-KS-KIN  PIC ¥¥¥,¥¥¥,¥¥¥.
           02    FILLER    PIC X(9)   VALUE    SPACE.
       01  LINE01.
           02    FILLER    PIC X(27)  VALUE    SPACE.
           02    FILLER    PIC X(12)  VALUE    ALL  "—".
           02    FILLER    PIC X(2)   VALUE    "┤".
           02    FILLER    PIC X(18)  VALUE    SPACE.
           02    FILLER    PIC X(12)  VALUE    ALL  "—".
           02    FILLER    PIC X(9)   VALUE    SPACE.
       01  LINE02.
           02    FILLER    PIC X(27)  VALUE    SPACE.
           02    FILLER    PIC X(12)  VALUE    ALL "=".
           02    FILLER    PIC X(2)   VALUE    "|".
           02    FILLER    PIC X(18)  VALUE    SPACE.
           02    FILLER    PIC X(12)  VALUE    ALL "=".
           02    FILLER    PIC X(9)   VALUE    SPACE.
       01  KANNAME         OCCURS  30.
           02    CODE      PIC 9(4).
           02    NAME      PIC X(16).
           02    FILLER    PIC X(2).
       01  HOZON.
           02    HOZONA    OCCURS  30.
             03  H-KR        PIC 9(4).
             03  H-KR-KIN    PIC 9(8).
             03  H-KS        PIC 9(4).
             03  H-KS-KIN    PIC 9(8).
       01  CRT-HYOUJI.
           02    CRT-NAME  PIC X(80)  VALUE "決算日を入力してください。".
           02    CRT-NEN   PIC X(80)  VALUE "年を入力[  ]".
           02    CRT-TUKI  PIC X(80)  VALUE "月を入力[  ]".
           02    CRT-HI    PIC X(80)  VALUE "日を入力[  ]".
       01  ENTER-SYORI.
           02    FILLER    PIC X(80)  VALUE    SPACE.
           02    FILLER    PIC X(9)   VALUE    SPACE.
           02    N-NEN     PIC 9(2).
           02    FILLER    PIC X(69)  VALUE    SPACE.
           02    FILLER    PIC X(9)   VALUE    SPACE.
           02    N-TUKI    PIC 9(2).
           02    FILLER    PIC X(69)  VALUE    SPACE.
           02    FILLER    PIC X(9)   VALUE    SPACE.
           02    N-HI      PIC 9(2).
           02    FILLER    PIC X(69)  VALUE    SPACE.
       PROCEDURE               DIVISION.
       HAJIMARI.
           OPEN   INPUT  KESSAN-F KANNAME-F  OUTPUT  INSATU-F.
           READ   KESSAN-F  AT  END  MOVE  1  TO  W-EOF01.
           PERFORM  SYORI01  UNTIL  W-EOF01  =  1.
       OWARI.
           CLOSE  KANNAME-F  KESSAN-F  INSATU-F.
           STOP   RUN.
       SYORI01.
           PERFORM  NYURYOKU.
           PERFORM  KAN-NAME   VARYING  WW  FROM  1  BY  1
```

```
                          UNTIL  WW  >  30  OR  1  =  W-EOF02.
      PERFORM  SYORI02    UNTIL  W-EOF01  =  1.
      PERFORM  MIDASI-INZI.
      PERFORM  RIEKI.
      PERFORM  INZI       VARYING  I   FROM  1  BY  1
                          UNTIL  I  NOT<  KR  AND  I  NOT<  KS.
      PERFORM  GOKEI-INZI.
 SYORI02.
      IF  K-CODE  >   7110
             IF  K-KR-ZAN  >  K-KS-ZAN
                   MOVE K-CODE TO  H-KR (KR)
                   MOVE K-KR-ZAN  TO  H-KR-KIN (KR)
                   ADD  K-KR-ZAN  TO  G-KR-KIN
                   ADD    1        TO  KR.
      IF  K-CODE  >   7110
             IF  K-KR-ZAN  <  K-KS-ZAN
                   MOVE K-CODE TO  H-KS (KS)
                   MOVE K-KS-ZAN  TO  H-KS-KIN (KS)
                   ADD  K-KS-ZAN  TO  G-KS-KIN
                   ADD    1        TO  KS.
      READ  KESSAN-F  AT  END  MOVE  1  TO  W-EOF01.
 SAGASI.
      IF  H-KR (I) =  CODE(T)  MOVE  NAME (T)  TO  O-KR.
      IF  H-KS (I) =  CODE(T)  MOVE  NAME (T)  TO  O-KS.
 MIDASI-INZI.
      WRITE   INSATU-REC  FROM  MIDASI01  AFTER  1.
      WRITE   INSATU-REC  FROM  MIDASI02  AFTER  3.
      WRITE   INSATU-REC  FROM  MIDASI03  AFTER  3.
 KAN-NAME.
      READ KANNAME-F INTO KANNAME (WW) AT END MOVE 1 TO W-EOF02.
 GOKEI-INZI.
      MOVE   SPACE   TO   O-KR  O-KS.
      WRITE   INSATU-REC  FROM  LINE01   AFTER  1.
      MOVE   G-KR-KIN  TO   O-KR-KIN.
      MOVE   G-KS-KIN  TO   O-KS-KIN.
      WRITE   INSATU-REC  FROM  OUT-R    AFTER  1.
      WRITE   INSATU-REC  FROM  LINE02   AFTER  1.
 INZI.
      MOVE   SPACE         TO  O-KR     O-KS.
      PERFORM  SAGASI  VARYING  T  FROM  1  BY  1
                          UNTIL  T  >  (WW - 1).
      MOVE  H-KR-KIN(I)   TO  O-KR-KIN.
      MOVE  H-KS-KIN(I)   TO  O-KS-KIN.
      WRITE   INSATU-REC  FROM  OUT-R    AFTER  1.
 NYURYOKU.
      DISPLAY  SPACE  UPON  CRT.
      DISPLAY  CRT-HYOUJI  AT  0102  UPON  CRT.
      DISPLAY  ENTER-SYORI   AT  0102  UPON  CRT.
      ACCEPT   ENTER-SYORI   AT  0102  FROM  CRT.
      MOVE   N-NEN   TO   O-NEN-01  O-NEN-02.
      MOVE   N-TUKI  TO   O-TUKI-01  O-TUKI-02.
      MOVE   N-HI    TO   O-HI-02.
 RIEKI.
      IF G-KR-KIN       >  G-KS-KIN
             COMPUTE H-KS-KIN (KS)  =  G-KR-KIN  -  G-KS-KIN
             MOVE   9998         TO  H-KS  (KS)
             MOVE   G-KR-KIN     TO  G-KS-KIN
             ADD      1          TO  KS.
      IF G-KR-KIN       <  G-KS-KIN
             COMPUTE H-KR-KIN (KR)  =  G-KS-KIN  -  G-KR-KIN
             MOVE   9999         TO  H-KR   (KR)
             MOVE   G-KS-KIN     TO  G-KR-KIN
             ADD      1          TO  KR.
```

(2) 実行結果

損　益　計　算　書

8年　3月　1日から　8年　3月　31日まで

自動車費	¥3,900	売上	¥84,010
給料	¥18,000		
雑費	¥6,705		
家賃	¥3,000		
燃料費	¥30,000		
当期純利益	¥22,405		
	¥84,010		¥84,010

Ⅲ　貸借対照表

(1) ソースプログラム

```
001010 IDENTIFICATION        DIVISION.
001020 PROGRAM-ID.           BS.
001030 AUTHOR.               武田　嘉孝(12345).
001040 ENVIRONMENT           DIVISION.
001050 CONFIGURATION         SECTION.
001060 SOURCE-COMPUTER.      PC-9801.
001070 OBJECT-COMPUTER.      PC-9801.
001080 SPECIAL-NAMES.        CONSOLE  IS  CRT.
001090 INPUT-OUTPUT          SECTION.
001100 FILE-CONTROL.
001110     SELECT KURIKOSI   ASSIGN "B:KURIKOSI.DAT".
001120     SELECT KANNAME-F  ASSIGN "B:KANJO.DAT".
```

```
            SELECT KESSAN-F    ASSIGN "B:KESSAN.DAT".
            SELECT INSATU-F    ASSIGN "PRN:".
       DATA                    DIVISION.
       FILE                    SECTION.
       FD   KANNAME-F.
       01   KANNAME-REC.
            02   FILLER        PIC X(22).
       FD   KURIKOSI.
       01   KURIKOSI-REC.
            02   I-NEN         PIC 9(2).
            02   I-TUKI        PIC 9(2).
            02   I-HI          PIC 9(2).
            02   I-BAN         PIC 9(4).
            02   I-KR          PIC 9(4).
            02   I-KS          PIC 9(4).
            02   I-KINGAKU     PIC 9(8).
            02   FILLER        PIC X(2).
       FD   KESSAN-F.
       01   KESSAN-REC.
            02   K-CODE        PIC 9(4).
            02   K-KR-ZAN      PIC 9(8).
            02   K-KS-ZAN      PIC 9(8).
            02   FILLER        PIC X(2).
       FD   INSATU-F  LABEL   RECORD  OMITTED.
       01   INSATU-REC.
            02   FILLER        PIC X(80).
       WORKING-STORAGE         SECTION.
       77   I                  PIC 9(2).
       77   T                  PIC 9(2).
       77   WW                 PIC 9(2).
       77   KR                 PIC 9(2) VALUE  1.
       77   KS                 PIC 9(2) VALUE  1.
       77   G-KR-KIN           PIC 9(8) VALUE  0.
       77   G-KS-KIN           PIC 9(8) VALUE  0.
       77   H-SA-KR            PIC 9(8) VALUE  0.
       77   H-SA-KS            PIC 9(8) VALUE  0.
       77   W-EOF01            PIC 9    VALUE  0.
       77   W-EOF02            PIC 9    VALUE  0.
       01   MIDASI01.
            02   FILLER    PIC X(30) VALUE   SPACE.
            02   FILLER    PIC X(50) VALUE "貸  借  対  照  表".
       01   MIDASI02.
            02   FILLER    PIC X(30) VALUE   SPACE.
            02   O-NEN     PIC Z(2).
            02   FILLER    PIC X(4)  VALUE "年".
            02   O-TUKI    PIC Z(2).
            02   FILLER    PIC X(4)  VALUE "月".
            02   O-HI      PIC Z(2).
            02   FILLER    PIC X(36) VALUE "日".
       01   MIDASI03.
            02   FILLER    PIC X(9)  VALUE   SPACE.
            02   FILLER    PIC X(30) VALUE   ALL  "―".
            02   FILLER    PIC X(2)  VALUE   "┬".
            02   FILLER    PIC X(30) VALUE   ALL  "―".
            02   FILLER    PIC X(9)  VALUE   SPACE.
       01   OUT-R.
            02   FILLER    PIC X(9)  VALUE   SPACE.
            02   O-KR      PIC X(16).
            02   FILLER    PIC X(4)  VALUE   SPACE.
            02   O-KR-KIN  PIC ¥¥,¥¥¥,¥¥¥.
            02   FILLER    PIC X(2)  VALUE   "|".
            02   O-KS      PIC X(16).
```

```
           02   FILLER    PIC X(3)  VALUE    SPACE.
           02   O-KS-KIN PIC ¥¥,¥¥¥,¥¥¥.
           02   FILLER    PIC X(9)  VALUE    SPACE.
       01  LINE01.
           02   FILLER    PIC X(27) VALUE    SPACE.
           02   FILLER    PIC X(12) VALUE    ALL  "─".
           02   FILLER    PIC X(2)  VALUE    "┤ ".
           02   FILLER    PIC X(18) VALUE    SPACE.
           02   FILLER    PIC X(12) VALUE    ALL  "─".
           02   FILLER    PIC X(9)  VALUE    SPACE.
       01  LINE02.
           02   FILLER    PIC X(27) VALUE    SPACE.
           02   FILLER    PIC X(12) VALUE    ALL "=".
           02   FILLER    PIC X(2)  VALUE    " | ".
           02   FILLER    PIC X(18) VALUE    SPACE.
           02   FILLER    PIC X(12) VALUE    ALL "=".
           02   FILLER    PIC X(9)  VALUE    SPACE.
       01  KANNAME        OCCURS  30.
           02   CODE      PIC 9(4).
           02   NAME      PIC X(16).
           02   FILLER    PIC X(2).
       01  HOZON.
           02   HOZONA    OCCURS  30.
             03  H-KR       PIC 9(4).
             03  H-KR-KIN   PIC 9(8).
             03  H-KS       PIC 9(4).
             03  H-KS-KIN   PIC 9(8).
       01  CRT-HYOUJI.
           02   CRT-NAME PIC X(80) VALUE "決算日を入力してください。".
           02   CRT-NEN  PIC X(80) VALUE "年を入力[  ]".
           02   CRT-TUKI PIC X(80) VALUE "月を入力[  ]".
           02   CRT-HI   PIC X(80) VALUE "日を入力[  ]".
       01  ENTER-SYORI.
           02   FILLER    PIC X(80) VALUE    SPACE.
           02   FILLER    PIC X(9)  VALUE    SPACE.
           02   N-NEN     PIC 9(2).
           02   FILLER    PIC X(69) VALUE    SPACE.
           02   FILLER    PIC X(9)  VALUE    SPACE.
           02   N-TUKI    PIC 9(2).
           02   FILLER    PIC X(69) VALUE    SPACE.
           02   FILLER    PIC X(9)  VALUE    SPACE.
           02   N-HI      PIC 9(2).
           02   FILLER    PIC X(69) VALUE    SPACE.
       PROCEDURE               DIVISION.
       HAJIMARI.
           OPEN INPUT  KESSAN-F KANNAME-F OUTPUT KURIKOSI INSATU-F.
           READ  KESSAN-F  AT  END  MOVE  1  TO  W-EOF01.
           PERFORM  SYORI01  UNTIL  W-EOF01  =  1.
       OWARI.
           CLOSE  KURIKOSI  KANNAME-F  KESSAN-F  INSATU-F.
           STOP   RUN.
       SYORI01.
           PERFORM  NYURYOKU.
           PERFORM  KAN-NAME    VARYING  WW  FROM  1  BY  1
                                UNTIL  WW  >  30  OR  1  =  W-EOF02.
           PERFORM  SYORI02     UNTIL  W-EOF01  =  1.
           PERFORM  MIDASI-INZI.
           PERFORM  RIEKI.
           PERFORM  INZI        VARYING  I   FROM  1  BY  1
                                UNTIL  I  NOT<  KR  AND  I  NOT<  KS.
           PERFORM  GOKEI-INZI.
           PERFORM  KAISI.
```

```
SYORI02.
     IF  K-CODE  NOT> 7110
             IF  K-KR-ZAN  >  K-KS-ZAN
                    MOVE K-CODE TO  H-KR (KR)
                    MOVE K-KR-ZAN  TO  H-KR-KIN (KR)
                    ADD  K-KR-ZAN  TO  G-KR-KIN
                    ADD    1       TO  KR.
     IF  K-CODE  NOT> 7110
             IF  K-KR-ZAN  <  K-KS-ZAN
                    MOVE K-CODE TO  H-KS (KS)
                    MOVE K-KS-ZAN  TO  H-KS-KIN (KS)
                    ADD  K-KS-ZAN  TO  G-KS-KIN
                    ADD    1       TO  KS.
     READ  KESSAN-F  AT  END  MOVE  1  TO  W-EOF01.
SAGASI.
     IF  H-KR (I) =  CODE(T)  MOVE  NAME (T)  TO  O-KR.
     IF  H-KS (I) =  CODE(T)  MOVE  NAME (T)  TO  O-KS.
MIDASI-INZI.
     WRITE    INSATU-REC  FROM  MIDASI01  AFTER  1.
     WRITE    INSATU-REC  FROM  MIDASI02  AFTER  3.
     WRITE    INSATU-REC  FROM  MIDASI03  AFTER  3.
KAN-NAME.
     READ KANNAME-F INTO KANNAME (WW) AT END MOVE 1 TO W-EOF02.
GOKEI-INZI.
     MOVE    SPACE   TO   O-KR  O-KS.
     WRITE    INSATU-REC  FROM  LINE01    AFTER  1.
     MOVE    G-KR-KIN  TO   O-KR-KIN.
     MOVE    G-KS-KIN  TO   O-KS-KIN.
     WRITE    INSATU-REC  FROM  OUT-R     AFTER  1.
     WRITE    INSATU-REC  FROM  LINE02    AFTER  1.
INZI.
     MOVE    SPACE          TO  O-KR      O-KS.
     PERFORM  SAGASI  VARYING  T  FROM  1  BY  1
                            UNTIL  T  >  (WW - 1).
     MOVE   H-KR-KIN(I)   TO  O-KR-KIN.
     MOVE   H-KS-KIN(I)   TO  O-KS-KIN.
     WRITE    INSATU-REC  FROM  OUT-R     AFTER  1.
NYURYOKU.
     DISPLAY  SPACE  UPON  CRT.
     DISPLAY  CRT-HYOUJI  AT  0102  UPON  CRT.
     DISPLAY  ENTER-SYORI   AT  0102  UPON  CRT.
     ACCEPT   ENTER-SYORI   AT  0102  FROM  CRT.
     MOVE    N-NEN    TO    O-NEN.
     MOVE    N-TUKI   TO    O-TUKI.
     MOVE    N-HI     TO    O-HI.
RIEKI.
     IF G-KR-KIN        >  G-KS-KIN
             COMPUTE H-SA-KS    =  G-KR-KIN  -  G-KS-KIN
             MOVE   H-SA-KS       TO  H-KS-KIN (KS)
             MOVE   9999          TO  H-KS    (KS)
             MOVE   G-KR-KIN      TO  G-KS-KIN
             ADD      1           TO  KS.
     IF G-KR-KIN        <  G-KS-KIN
             COMPUTE H-SA-KR    =  G-KS-KIN  -  G-KR-KIN
             MOVE   H-SA-KR       TO  H-KR-KIN (KR)
             MOVE   9998          TO  H-KR    (KR)
             MOVE   G-KS-KIN      TO  G-KR-KIN
             ADD      1           TO  KR.
KAISI.
     MOVE "開 始 仕 訳 デ ー タ"  TO  INSATU-REC.
     WRITE    INSATU-REC    AFTER  3
     MOVE    SPACE       TO  KURIKOSI-REC.
```

```
           ADD       1         TO   N-TUKI.
           MOVE   0001         TO   I-BAN.
           MOVE   N-NEN        TO   I-NEN.
           MOVE   N-TUKI       TO   I-TUKI.
           IF  N-TUKI  NOT< 13
               MOVE     01     TO   I-TUKI
               ADD      1      TO   I-NEN.
               MOVE     1      TO   I-HI.
           PERFORM   KAISI-KR  VARYING  I  FROM  1  BY  1
                                 UNTIL  I  NOT<  KR.
           PERFORM   KAISI-KS  VARYING  I  FROM  1  BY  1
                                 UNTIL  I  NOT<  KS.
       KAISI-KR.
           MOVE   9999         TO   I-KR   I-KS.
           IF H-KR  (I)   NOT=  9998
                  MOVE   H-KR(I)         TO   I-KR
                  MOVE   H-KR-KIN(I)  TO   I-KINGAKU
                  WRITE  INSATU-REC  FROM  KURIKOSI-REC  AFTER  1
                  WRITE  KURIKOSI-REC  AFTER  1.
       KAISI-KS.
           MOVE   9999         TO   I-KR   I-KS.
           IF H-KS  (I)   NOT=  9999  AND  H-KS (I)  NOT= 7110
                  MOVE   H-KS(I)         TO   I-KS
                  MOVE   H-KS-KIN(I)  TO   I-KINGAKU
                  WRITE  INSATU-REC  FROM  KURIKOSI-REC  AFTER  1
                  WRITE  KURIKOSI-REC.
           IF H-KS  (I)       =   7110
                  COMPUTE I-KINGAKU = H-KS-KIN(I) + H-SA-KS - H-SA-KR
                  MOVE   7110            TO   I-KS
                  WRITE  INSATU-REC  FROM  KURIKOSI-REC  AFTER  1
                  WRITE  KURIKOSI-REC  AFTER  1.
```

(2) 実行結果

① 貸借対照表

② 開始仕訳データ

③ 繰越データ

```
                    貸　借　対　照　表

                      8年　 3月　31日

─────────────────────┬────────────────────────
現金              ¥23,155 │ 買掛金              ¥13,120
売掛金            ¥51,600 │ 資本金             ¥303,610
車両運搬具       ¥239,960 │ 当期純利益          ¥22,405
備品              ¥24,420 │
           ───────────┤                 ─────────
                 ¥339,135 │                   ¥339,135
                 ======== │                   ========

開始仕訳データ
0804010001111199990002315５
08040100011131999900051600
08040100012151999900239960
08040100012161999900024420
08040100019999412000013120
08040100019999711000326015

B:¥KURIKOSI.DAT                    [   7:   1]     <挿入>          HELP=･
----+----1----+----2----+----3----+----4----+----5----+----6----+----7----
08040100011111999900023155
08040100011131999900051600
08040100012151999900239960
08040100012161999900024420
08040100019999412000013120
08040100019999711000326015
[EOF]
~
~
~
~
~
~
~
~
```

Ⅳ　繰越試算表

(1)　ソースプログラム

```
IDENTIFICATION             DIVISION.
PROGRAM-ID.                KURITB.
AUTHOR.                    武田　嘉孝(12345).
ENVIRONMENT                DIVISION.
CONFIGURATION              SECTION.
SOURCE-COMPUTER.           PC-9801.
OBJECT-COMPUTER.           PC-9801.
SPECIAL-NAMES.             CONSOLE  IS  CRT.
INPUT-OUTPUT               SECTION.
FILE-CONTROL.
     SELECT KURIKOSI       ASSIGN "B:KURIKOSI.DAT".
     SELECT KANNAME-F      ASSIGN "B:KANJO.DAT".
     SELECT KESSAN-F       ASSIGN "B:KESSAN.DAT".
     SELECT INSATU-F       ASSIGN "PRN:".
DATA                       DIVISION.
FILE                       SECTION.
FD   KANNAME-F.
01   KANNAME-REC.
     02    FILLER          PIC X(22).
FD   KURIKOSI.
01   KURIKOSI-REC.
     02    I-NEN           PIC 9(2).
     02    I-TUKI          PIC 9(2).
     02    I-HI            PIC 9(2).
     02    I-BAN           PIC 9(4).
     02    I-KR            PIC 9(4).
     02    I-KS            PIC 9(4).
     02    I-KINGAKU       PIC 9(8).
     02    FILLER          PIC X(2).
FD   KESSAN-F.
01   KESSAN-REC.
     02    K-CODE          PIC 9(4).
     02    K-KR-ZAN        PIC 9(8).
     02    K-KS-ZAN        PIC 9(8).
     02    FILLER          PIC X(2).
FD   INSATU-F  LABEL  RECORD  OMITTED.
01   INSATU-REC.
     02    FILLER          PIC X(80).
WORKING-STORAGE            SECTION.
77   I                     PIC 9(2).
77   T                     PIC 9(2).
77   WW                    PIC 9(2).
77   KR                    PIC 9(2) VALUE  1.
77   KS                    PIC 9(2) VALUE  1.
77   G-KR-KIN              PIC 9(8) VALUE  0.
77   G-KS-KIN              PIC 9(8) VALUE  0.
77   H-SA-KR               PIC 9(8) VALUE  0.
77   H-SA-KS               PIC 9(8) VALUE  0.
77   W-EOF01               PIC 9    VALUE  0.
77   W-EOF02               PIC 9    VALUE  0.
01   MIDASI01.
     02    FILLER     PIC X(30) VALUE   SPACE.
     02    FILLER     PIC X(50) VALUE "繰　越　試　算　表".
01   MIDASI02.
     02    FILLER     PIC X(30) VALUE   SPACE.
     02    O-NEN      PIC Z(2).
     02    FILLER     PIC X(4)  VALUE "年".
     02    O-TUKI     PIC Z(2).
```

```
       02   FILLER    PIC X(4)   VALUE "月".
       02   O-HI      PIC Z(2).
       02   FILLER    PIC X(36)  VALUE "日".
01     MIDASI03.
       02   FILLER    PIC X(9)   VALUE   SPACE.
       02   FILLER    PIC X(30)  VALUE   ALL  "─".
       02   FILLER    PIC X(2)   VALUE   "┬".
       02   FILLER    PIC X(30)  VALUE   ALL  "─".
       02   FILLER    PIC X(9)   VALUE   SPACE.
01     OUT-R.
       02   FILLER    PIC X(9)   VALUE   SPACE.
       02   O-KR      PIC X(16).
       02   FILLER    PIC X(4)   VALUE   SPACE.
       02   O-KR-KIN  PIC ¥¥,¥¥¥,¥¥¥.
       02   FILLER    PIC X(2)   VALUE   "│".
       02   O-KS      PIC X(16).
       02   FILLER    PIC X(3)   VALUE   SPACE.
       02   O-KS-KIN  PIC ¥¥,¥¥¥,¥¥¥.
       02   FILLER    PIC X(9)   VALUE   SPACE.
01     LINE01.
       02   FILLER    PIC X(27)  VALUE   SPACE.
       02   FILLER    PIC X(12)  VALUE   ALL  "─".
       02   FILLER    PIC X(2)   VALUE   "┤".
       02   FILLER    PIC X(18)  VALUE   SPACE.
       02   FILLER    PIC X(12)  VALUE   ALL  "─".
       02   FILLER    PIC X(9)   VALUE   SPACE.
01     LINE02.
       02   FILLER    PIC X(27)  VALUE   SPACE.
       02   FILLER    PIC X(12)  VALUE   ALL "＝".
       02   FILLER    PIC X(2)   VALUE   "│".
       02   FILLER    PIC X(18)  VALUE   SPACE.
       02   FILLER    PIC X(12)  VALUE   ALL "＝".
       02   FILLER    PIC X(9)   VALUE   SPACE.
01     KANNAME        OCCURS  30.
       02   CODE      PIC 9(4).
       02   NAME      PIC X(16).
       02   FILLER    PIC X(2).
01     HOZON.
       02   HOZONA    OCCURS  30.
         03  H-KR        PIC 9(4).
         03  H-KR-KIN    PIC 9(8).
         03  H-KS        PIC 9(4).
         03  H-KS-KIN    PIC 9(8).
01     CRT-HYOUJI.
       02   CRT-NAME PIC X(80) VALUE "決算日を入力してください。".
       02   CRT-NEN  PIC X(80) VALUE "年を入力[  ]".
       02   CRT-TUKI PIC X(80) VALUE "月を入力[  ]".
       02   CRT-HI   PIC X(80) VALUE "日を入力[  ]".
01     ENTER-SYORI.
       02   FILLER    PIC X(80) VALUE   SPACE.
       02   FILLER    PIC X(9)  VALUE   SPACE.
       02   N-NEN     PIC 9(2).
       02   FILLER    PIC X(69) VALUE   SPACE.
       02   FILLER    PIC X(9)  VALUE   SPACE.
       02   N-TUKI    PIC 9(2).
       02   FILLER    PIC X(69) VALUE   SPACE.
       02   FILLER    PIC X(9)  VALUE   SPACE.
       02   N-HI      PIC 9(2).
       02   FILLER    PIC X(69) VALUE   SPACE.
PROCEDURE             DIVISION.
HAJIMARI.
       OPEN INPUT  KESSAN-F KANNAME-F OUTPUT KURIKOSI INSATU-F.
```

```
     READ  KESSAN-F   AT  END  MOVE  1  TO  W-EOF01.
     PERFORM  SYORI01  UNTIL  W-EOF01  =  1.
OWARI.
     CLOSE  KURIKOSI  KANNAME-F  KESSAN-F  INSATU-F.
     STOP   RUN.
SYORI01.
     PERFORM  NYURYOKU.
     PERFORM  KAN-NAME    VARYING  WW  FROM  1  BY  1
                          UNTIL  WW  >  30  OR  1  =  W-EOF02.
     PERFORM  SYORI02     UNTIL  W-EOF01  =  1.
     PERFORM  MIDASI-INZI.
     PERFORM  INZI        VARYING  I   FROM  1  BY  1
                          UNTIL  I  NOT<  KR  AND  I  NOT< KS.
     PERFORM  GOKEI-INZI.
     PERFORM  KAISI.
SYORI02.
     IF  K-CODE    <   7110
             IF  K-KR-ZAN  >  K-KS-ZAN
                   MOVE K-CODE TO  H-KR (KR)
                   MOVE K-KR-ZAN  TO  H-KR-KIN (KR)
                   ADD  K-KR-ZAN  TO  G-KR-KIN
                   ADD    1       TO  KR.
     IF  K-CODE  <    7110
             IF  K-KR-ZAN  <  K-KS-ZAN
                   MOVE K-CODE TO  H-KS (KS)
                   MOVE K-KS-ZAN  TO  H-KS-KIN (KS)
                   ADD  K-KS-ZAN  TO  G-KS-KIN
                   ADD    1       TO  KS.
     IF  K-CODE  =    7110
             IF  G-KR-KIN  >  G-KS-KIN
                   MOVE K-CODE TO  H-KS(KS)
                   COMPUTE H-SA-KR = G-KR-KIN - G-KS-KIN
                   MOVE H-SA-KR   TO  H-KS-KIN(KS)
                   ADD  H-SA-KR   TO  G-KS-KIN
                   ADD    1       TO  KS.
     IF  K-CODE  =    7110
             IF  G-KR-KIN  <  G-KS-KIN
                   MOVE K-CODE TO  H-KS(KS)
                   COMPUTE H-SA-KS = G-KS-KIN - G-KR-KIN
                   COMPUTE H-SA-KS  = K-KS-ZAN - H-SA-KS
                   MOVE H-SA-KS   TO  H-KS-KIN(KS)
                   ADD  H-SA-KS   TO  G-KR-KIN
                   ADD    1       TO  KS.
     READ  KESSAN-F AT  END  MOVE  1  TO W-EOF01.
SAGASI.
     IF  H-KR (I) =  CODE(T)  MOVE  NAME (T)  TO  O-KR.
     IF  H-KS (I) =  CODE(T)  MOVE  NAME (T)  TO  O-KS.
MIDASI-INZI.
     WRITE    INSATU-REC  FROM  MIDASI01  AFTER  1.
     WRITE    INSATU-REC  FROM  MIDASI02  AFTER  3.
     WRITE    INSATU-REC  FROM  MIDASI03  AFTER  3.
KAN-NAME.
     READ KANNAME-F INTO KANNAME (WW) AT END MOVE 1 TO W-EOF02.
GOKEI-INZI.
     MOVE   SPACE   TO   O-KR  O-KS.
     WRITE    INSATU-REC  FROM  LINE01   AFTER  1.
     MOVE   G-KR-KIN  TO   O-KR-KIN.
     MOVE   G-KS-KIN  TO   O-KS-KIN.
     WRITE    INSATU-REC  FROM  OUT-R    AFTER  1.
     WRITE    INSATU-REC  FROM  LINE02   AFTER  1.
INZI.
     MOVE   SPACE          TO  O-KR      O-KS.
     PERFORM  SAGASI   VARYING  T  FROM  1  BY  1
```

```
                              UNTIL  T  >  (WW - 1).
       MOVE  H-KR-KIN(I)    TO  O-KR-KIN.
       MOVE  H-KS-KIN(I)    TO  O-KS-KIN.
       WRITE    INSATU-REC  FROM  OUT-R    AFTER  1.
 NYURYOKU.
       DISPLAY  SPACE  UPON  CRT.
       DISPLAY  CRT-HYOUJI  AT  0102  UPON  CRT.
       DISPLAY  ENTER-SYORI   AT  0102  UPON  CRT.
       ACCEPT   ENTER-SYORI   AT  0102  FROM  CRT.
       MOVE    N-NEN   TO    O-NEN.
       MOVE    N-TUKI  TO    O-TUKI.
       MOVE    N-HI    TO    O-HI.
 KAISI.
       MOVE "開  始  仕  訳  デ  ー  タ"  TO  INSATU-REC.
       WRITE INSATU-REC  AFTER  4.
       MOVE    SPACE   TO  KURIKOSI-REC.
       ADD     1       TO  N-TUKI.
       MOVE    0001    TO  I-BAN.
       MOVE    N-NEN   TO  I-NEN.
       MOVE    N-TUKI  TO  I-TUKI.
       IF      N-TUKI NOT< 13
               MOVE    01  TO  I-TUKI
               ADD  1  TO  I-NEN.
       MOVE    1       TO  I-HI.
       PERFORM    KAISI-KR   VARYING I FROM 1  BY 1
                             UNTIL  I  NOT<  KR.
       PERFORM    KAISI-KS   VARYING I FROM 1  BY 1
                             UNTIL  I  NOT<  KS.
 KAISI-KR.
       MOVE    9999    TO  I-KR  I-KS.
       IF  H-KR  (I)  NOT=  9998
               MOVE   H-KR(I)        TO  I-KR
               MOVE   H-KR-KIN(I)    TO  I-KINGAKU
               WRITE  INSATU-REC   FROM KURIKOSI-REC AFTER 1
               WRITE  KURIKOSI-REC  AFTER  1.
 KAISI-KS.
       MOVE    9999    TO  I-KR  I-KS.
       IF  H-KS  (I)  NOT=  9999  AND  H-KS (I)  NOT= 7110
               MOVE   H-KS(I)        TO  I-KS
               MOVE   H-KS-KIN(I)    TO  I-KINGAKU
               WRITE  INSATU-REC   FROM KURIKOSI-REC AFTER 1
               WRITE  KURIKOSI-REC  AFTER  1.
       IF  H-KS  (I)  =    7110
               MOVE   H-KS-KIN(I)    TO  I-KINGAKU
               MOVE      7110        TO  I-KS
               WRITE  INSATU-REC   FROM KURIKOSI-REC AFTER 1
               WRITE  KURIKOSI-REC  AFTER  1.
```

(2) 実行結果

① 繰越試算表

② 開始仕訳データ

繰　越　試　算　表

8年　3月　31日

現金	¥23,155	買掛金	¥13,120
売掛金	¥51,600	資本金	¥326,015
車両運搬具	¥239,960		
備品	¥24,420		
	¥339,135		¥339,135

開　始　仕　訳　デ　ー　タ

```
08040100011111999900023155
08040100011131999900051600
08040100012151999900239960
08040100012161999900024420
08040100019999412000013120
08040100019999711000326015
```

おわりに

本稿では，簿記原理で取り扱われる重要な帳簿である(1)合計残高試算表，(2)損益計算書，(3)貸借対照表，(4)繰越試算表をコンピュータ・プログラミングを通じて，その作成方法を提示した。本稿分を含めて筆者が作成したパソコンコボルプログラムによる帳簿には，次のものがある。(1)仕訳データ入力帳，(2)仕訳帳（二種類），(3)総勘定元帳（仕訳帳兼元帳），(4)合計残高試算表（二種類），(5)損益計算書，(6)貸借対照表，(7)繰越試算表，それに，(8)データ形式変換プログラム，(9)仕訳データソートプログラム（二種類），(10)コンピュータ会計メニュー画面プログラム（二種類）

これらは全て，マイクロソフト社のディスク・オペレイティング・システム（MS-DOS）上でCOBOLを用いて作成したものである。

ここ最近になり，Windows 3.1と95がOSの主流となってきている。筆者もWindows 3.1については，富士通FMV-DESKPOWERを用い，Windows用のMF-COBOL（マイクロフォーカス社のコボル）とMIFESのエディタによって，すでにプログラムの開発を手掛けている。コンパイル＆ゴーの方法として，このWindows 3.1の場合，2つのアプローチを試みた。1つは，MS-DOS 6.2Vからそれを行い，実行結果を画面に表示してから確認し，Windows 3.1に切り換えてから，アクセサリをダブルクリックし，ライトもダブルクリックし，その後ファイルを開いてソースプログラムの内容を画面に表示したり，実行結果のファイルを同じ方法で画面に表示させ，それぞれ印刷させる方法である。他の1つは，Windows 3.1から起動して，MIFESのエディタをダブルクリックし，データ及びプログラムを入力し，セーブし，その後コンパイル＆ゴーを行い，アクセサリー，ライトをそれぞれ順序よくダブルクリックしてから，実行結果のファイル，及びソースプログラムのファイルを開いて，それぞれ印刷させる方法である。この二つのアプローチのいずれの場合でも，従来のMS-DOSがキーボードに付いている COPY ボタンを押すことによって画面印字（ハードコピー）ができたのに対して，Windows 3.1ではその機能が備えられていないということである。これは，プリンタデバイス（レーザープリンタ）との関係から生ずる1つの問題点であると思う。しかし、その主たる原因はWindows 3.1でも95でも、その機能は備えていないと考えた方が良い。但し、ハードコピーをどうしても行うためには、COPY ボタン（ Print Screen ボタン）を押し、ペイントブラシに貼り付ける方法がある。その後でファイルと印刷をクリックして画面印字をするやり方でしか、ハードコピーは不可能である。

またWindows 95では，NEC PC-9821を用い，MS-DOSプロンプトからCOBOLプログラムとデータを入力し，セーブをそれぞれ行ってからコンパイル＆ゴーを試みた。このアプローチでは，MS-DOSプロンプトの画面で表示された実行結果のデータを，ファイルに格納し

なくても，ワードパッド（又はメモ帳）に貼り付けができ，そして文字の大きさはポイントを選択指定して，変更ができる。（Windows 3.1のライトでも文字の大きさは変更できる。）但し，メモ帳ではそれができない。またハードコピーは，Windowsの場合，直接レーザープリンタにプリントアウトすることは不可能であった。

このようにWindows 95は，Windows 3.1に比べると，確かに機能性が向上しているし，操作の手間が省略され，利便性をもって真にグレードアップされている感じをもつ。その実例の一つとして，MS-DOSプロンプトの充実度を挙げることができる。この一例からもわかるように，Windows 3.1における物足りなさが，Windows 95によって補足されているようである。今後は，両方のWindowsの特徴を知り，それを生かしていくためにも，２つのハードディスク（内蔵と外付け）にドライブを分けてOSを格納し，併用しようと考えている。その目的は，Windows環境をフルに活用し，MF-COBOLによって各種の会計帳簿の作成を試みるためである。

参　考　文　献

1）武田安弘著『基礎会計学』（改訂版）税務経理協会，1996年３月

2）武田安弘著『現代会計学入門』（新訂版）税務経理協会，1995年７月

3）武田嘉孝著「第22章 帳簿組織」中村萬次，津谷原弘編，『新訂　複式簿記の構造』同文舘，1996年４月，265－287頁

4）武田嘉孝著「仕訳月報作成のためのコンピュータ処理」，聖徳学園女子短期大学紀要第26集，1996年３月，89－101頁

5）武田嘉孝著「コボルによる会計データ処理の一研究」，聖徳学園女子短期大学紀要第24集，1995年３月，105－139頁

6）武田嘉孝著『コンピュータ会計論』同文舘，1994年５月

7）武田嘉孝編著『会計情報処理』中日教育会，1992年７月